
VAR
Fp : ARRAY[ 0..1 ] OF FILEP; (* File pointer *)
ABf : BOOL; (* Auxiliary buffer *)
Sm : SYSSERIALMODE; (* Serial mode *)
MMdb : ModbusMaster; (* Modbus master communication *)
MIDx : USINT; (* Modbus index *)
SktLsn : SysSktListen; (* FB Socket listen *)
SData : STRING[ 260 ]; (* Socket data *)
DExch : DataStreamExch; (* Data stream exchange *)
MdbENr : UINT; (* Modbus error number *)
MdbCfg : ARRAY[ 0..8 ] OF MODBUSCONFIG; (* Modbus configuration *)
MdbValue : UDINT; (* Valore per modbus *)
FMIDx : USINT; (* Force Modbus index *)
TimeChk : ARRAY[ 0..2 ] OF UDINT; (* Controllo tempo *)
END_VAR

   1 (* ************************************************************************** *)
   2 (* PROGRAM "MasterModbus"                                                     *)
   3 (* ************************************************************************** *)
   4 (* Questo programma gestisce la comunicazione modbus con i vari slaves.       *)
   5 (* -------------------------------------------------------------------------- *)
   6 (* Eseguo inizializzazioni. *)
   7
   8 IF (SysFirstLoop) THEN
   9
  10 (* ------------------------------------------------------------------ *)
  11 (* CONFIGURAZIONE REGISTRI MODBUS                                     *)
  12 (* ------------------------------------------------------------------ *)
  13 (* Scrittura "Control Word". *)
  14
  15 MdbCfg[0].Node:=1; (* Modbus node *)
  16 MdbCfg[0].FCode:=16#10; (* Modbus function *)
  17 MdbCfg[0].Address:=16#8000; (* Register address *)
  18 MdbCfg[0].Buffer:=ADR(PMandrino.ControlWord); (* Buffer address *)
  19
  20 (* Scrittura "Modes Of Operation". *)
  21
  22 MdbCfg[1].Node:=1; (* Modbus node *)
  23 MdbCfg[1].FCode:=16#10; (* Modbus function *)
  24 MdbCfg[1].Address:=16#8001; (* Register address *)
  25 MdbCfg[1].Buffer:=ADR(PMandrino.MOperation); (* Buffer address *)
  26
  27 (* Lettura "Status Word". *)
  28
  29 MdbCfg[2].Node:=1; (* Modbus node *)
  30 MdbCfg[2].FCode:=16#03; (* Modbus function *)
  31 MdbCfg[2].Address:=16#8005; (* Register address *)
  32 MdbCfg[2].Buffer:=ADR(PMandrino.StatusWord); (* Buffer address *)
  33
  34 (* Lettura "Position Actual Value". *)
  35
  36 MdbCfg[3].Node:=1; (* Modbus node *)
  37 MdbCfg[3].FCode:=16#03; (* Modbus function *)
  38 MdbCfg[3].Address:=16#8006; (* Register address *)
  39 MdbCfg[3].Buffer:=ADR(PMandrino.PositionAV); (* Buffer address *)
  40
  41 (* Scrittura "Target position". *)

  42
  43 MdbCfg[4].Node:=1; (* Modbus node *)
  44 MdbCfg[4].FCode:=16#10; (* Modbus function *)
  45 MdbCfg[4].Address:=16#8009; (* Register address *)
  46 MdbCfg[4].Buffer:=ADR(PMandrino.TPosition); (* Buffer address *)
  47
  48 (* Scrittura "Profile velocity". *)
  49
  50 MdbCfg[5].Node:=1; (* Modbus node *)
  51 MdbCfg[5].FCode:=16#10; (* Modbus function *)
  52 MdbCfg[5].Address:=16#800C; (* Register address *)
  53 MdbCfg[5].Buffer:=ADR(PMandrino.PVelocity); (* Buffer address *)
  54
  55 (* Scrittura "Homing method". *)
  56
  57 MdbCfg[6].Node:=1; (* Modbus node *)
  58 MdbCfg[6].FCode:=16#10; (* Modbus function *)
  59 MdbCfg[6].Address:=16#8012; (* Register address *)
  60 MdbCfg[6].Buffer:=ADR(PMandrino.HMethod); (* Buffer address *)
  61
  62 (* Scrittura "Homing fast speed". *)
  63
  64 MdbCfg[7].Node:=1; (* Modbus node *)
  65 MdbCfg[7].FCode:=16#10; (* Modbus function *)
  66 MdbCfg[7].Address:=16#8014; (* Register address *)
  67 MdbCfg[7].Buffer:=ADR(PMandrino.HFastSpeed); (* Buffer address *)
  68
  69 (* Scrittura "Homing accelleration". *)
  70
  71 MdbCfg[8].Node:=1; (* Modbus node *)
  72 MdbCfg[8].FCode:=16#10; (* Modbus function *)
  73 MdbCfg[8].Address:=16#8016; (* Register address *)
  74 MdbCfg[8].Buffer:=ADR(PMandrino.HAcc); (* Buffer address *)
  75
  76 (* ------------------------------------------------------------------ *)
  77 (* APERTURA STREAMS DI COMUNICAZIONE                                  *)
  78 (* ------------------------------------------------------------------ *)
  79 (* Apro ed inizializzo porta seriale. *)
  80
  81 Fp[0]:=Sysfopen('COM2', 'rw'); (* File pointer *)
  82 ABf:=SysGetSerialMode(ADR(Sm), Fp[0]); (* Auxiliary buffer *)
  83 Sm.Baudrate:=19200; (* Baud rate *)
  84 Sm.Parity:='E'; (* Parity *)
  85 Sm.DataBits:=8; (* Data bits *)
  86 Sm.DTRManagement:=DTR_AUTO_WO_TIMES; (* DTR management *)
  87 ABf:=SysSetSerialMode(ADR(Sm), Fp[0]); (* Auxiliary buffer *)
  88
  89 (* Eseguo apertura socket. *) 
  90
  91 Fp[1]:=Sysfopen('TCPSKT', 'rw'); (* File pointer *)
  92
  93 (* ------------------------------------------------------------------ *)
  94 (* PARAMETRIZZAZIONE FBs                                              *)
  95 (* ------------------------------------------------------------------ *)
  96 (* Inizializzo FB comunicazione modbus. *)
  97
  98 MIDx:=0; (* Modbus index *)
  99 MMdb.File:=Fp[0]; (* File pointer *)
 100 MMdb.IFTime:=1720; (* Tempo tra frames *)
 101 MMdb.Timeout:=100; (* Communication timeout *)

 102 MMdb.Delay:=10; (* Communication delay *)
 103
 104 (* Inizializzo FB scambio dati tra socket e seriale. *)
 105
 106 DExch.FpA:=Fp[0]; (* File pointer (Stream A) *)
 107 DExch.FpB:=Fp[1]; (* File pointer (Stream B) *)
 108 DExch.DBSize:=SIZEOF(SData); (* Data buffer size *)
 109 DExch.DDelay:=30; (* Data delay *)
 110 END_IF;
 111
 112 (* ---------------------------------------------------------------------- *)
 113 (* GESTIONE CONTROLLO STATO SOCKET TCP                                    *)
 114 (* ---------------------------------------------------------------------- *)
 115 (* Forzo socket in condizione di listening. *)
 116 (* "LifeTm" piccolo, forza chiusura socket su mancanza comunicazione. *)
 117 (* "FlushTm" deve essere più grande del tempo di loop programma. *)
 118
 119 SktLsn.File:=Fp[1]; (* Flusso dati stream *)
 120 SktLsn.MyPort:=2000; (* Porta in ascolto su socket *)
 121 SktLsn.LifeTm:=60; (* Tempo di vita socket (S) *)
 122 SktLsn.FlushTm:=10; (* Tempo di flush socket (mS) *)
 123 SktLsn.RxSize:=TO_UINT(SIZEOF(SData)); (* Dimensione buffer Rx dati socket *)
 124 SktLsn.TxSize:=TO_UINT(SIZEOF(SData)); (* Dimensione buffer Tx dati socket *)
 125 SktLsn(Enable:=TRUE); (* Forzo socket in listening *)
 126
 127 (* ---------------------------------------------------------------------- *)
 128 (* GESTIONE SCAMBIO DATI TRA SOCKET TCP E SERIALE                         *)
 129 (* ---------------------------------------------------------------------- *)
 130 (* Se client è connesso al socket non viene più eseguita la comunicazione *)
 131 (* modbus con i nodi slaves ora i dati arrivano dalla connessione TCP/IP. *)
 132
 133 DExch(Enable:=SktLsn.Connect); (* Data stream exchange *)
 134 IF (SktLsn.Connect) THEN MIDx:=0; RETURN; END_IF;
 135
 136 (* ---------------------------------------------------------------------- *)
 137 (* GESTIONE COMUNICAZIONE MODBUS                                          *)
 138 (* ---------------------------------------------------------------------- *)
 139 (* Eseguo controllo e conteggio errori di comunicazione. *)
 140
 141 IF (MMdb.Fault) THEN MdbENr:=MdbENr+1; END_IF;
 142
 143 (* Eseguo controllo se comando Modbus eseguito correttamente. *)
 144
 145 IF (MMdb.Ok) THEN
 146
 147 (* Se comando lettura eseguo swap valore letto. *)
 148
 149 IF (MdbCfg[MIDx].FCode = 16#03) THEN MdbCfg[MIDx].@Buffer:=ROL(MdbValue, 16); END_IF;
 150
 151 (* Trasferisco valore in "Value", in questo modo il comando di *)
 152 (* scrittura non verrà più eseguito. Copio il reale valore inviato *)
 153 (* in questo modo sono al riparo da variazioni in task Fast e Slow. *)
 154
 155 MdbCfg[MIDx].Value:=ROL(MdbValue, 16); (* Valore parametro *)
 156 END_IF;
 157
 158 (* Su fine esecuzione comando Modbus passo a gestire nuovo nodo. *)
 159
 160 IF (MMdb.Done) THEN
 161

 162 (* Eseguo incremento indice tabella configurazione Modbus. *)
 163
 164 MIDx:=MIDx+1; (* Modbus index *)
 165 IF (MIDx >= (SIZEOF(MdbCfg)/SIZEOF(MdbCfg[0]))) THEN
 166
 167 TimeChk[0]:=SysGetSysTime(TRUE)-TimeChk[1];
 168 TimeChk[1]:=SysGetSysTime(TRUE);
 169 IF (TimeChk[0] > TimeChk[2]) THEN TimeChk[2]:=TimeChk[0]; END_IF;
 170
 171 (* Azzero indice modbus ed incremento indice forzatura. *)
 172
 173  MIDx:=0; (* Modbus index *)
 174 FMIDx:=FMIDx+1; (* Force Modbus index *)
 175 IF (FMIDx >= (SIZEOF(MdbCfg)/SIZEOF(MdbCfg[0]))) THEN FMIDx:=0; END_IF;
 176 END_IF;
 177
 178 (* Su comando scrittura Modbus controllo se valore già scritto. *)
 179
 180 IF (MdbCfg[MIDx].FCode = 16#10) THEN
 181 IF (MIDx <> FMIDx) THEN
 182 IF (MdbCfg[MIDx].Value = MdbCfg[MIDx].@Buffer) THEN RETURN; END_IF;
 183 END_IF;
 184
 185 (* Eseguo swap valore da scrivere verso Modbus. *)
 186
 187 MdbValue:=ROL(MdbCfg[MIDx].@Buffer, 16); (* Valore per modbus *)
 188 END_IF;
 189
 190 (* Eseguo reset abilitazione FB per eseguire nuovo comando. *)
 191
 192 MMdb.Enable:=FALSE; (* FB enable *)
 193 END_IF;
 194
 195 (* Gestione comunicazione modbus. *)
 196
 197 MMdb(); (* Modbus master communication *)
 198
 199 (* Eseguo parametrizzazione FB modbus, sommo 1 per offset Modbus. *)
 200 (* Tutti i comandi operano su doppio registro (16 bits). *)
 201
 202 IF NOT(MMdb.Enable) THEN
 203 MMdb.Type:=0; (* Modbus type *)
 204 MMdb.Address:=MdbCfg[MIDx].Address+1; (* Variable address *)
 205 MMdb.Buffer:=ADR(MdbValue); (* Buffer address *)
 206 MMdb.Points:=2; (* Number of points *)
 207 MMdb.Node:=MdbCfg[MIDx].Node; (* Modbus node *)
 208 MMdb.FCode:=MdbCfg[MIDx].FCode; (* Modbus function *)
 209 MMdb.Enable:=TRUE; (* FB enable *)
 210 END_IF;
 211
 212 (* [End of file] *)
 213
 214


